Dirac equations

Let us introduce the following Ore algebra A of partial differential operators

```
D1 = Der[x1]; D2 = Der[x2]; D3 = Der[x3]; D4 = Der[x4];
A = OreAlgebra[D1, D2, D3, D4]
```

$\mathbb{K}(x1, x2, x3, x4)[D_{x1}; 1, D_{x1}][D_{x2}; 1, D_{x2}][D_{x3}; 1, D_{x3}][D_{x4}; 1, D_{x4}]$

and the matrix R of partial differential operators which defines the Dirac equations for massless particule

```
MatrixForm[R = ToOrePolynomial[{{D4, 0, -I D3, -I D1 - D2},
    {0, D4, -I D1 + D2, I D3},
    {I D3, I D1 + D2, -D4, 0},
    {I D1 - D2, -I D3, 0, -D4}}, A]]
```

$$\begin{pmatrix} D_{x4} & 0 & -i\,D_{x3} & -i\,D_{x1} - D_{x2} \\ 0 & D_{x4} & -i\,D_{x1} + D_{x2} & i\,D_{x3} \\ i\,D_{x3} & i\,D_{x1} + D_{x2} & -D_{x4} & 0 \\ i\,D_{x1} - D_{x2} & -i\,D_{x3} & 0 & -D_{x4} \end{pmatrix}$$

and the left A-module, finitely presented by R, defined by M = $A^{1\times 4} / A^{1\times 4} R$.

Let us check whether or not M can be decomposed. We first compute the endomorphisms of M defined a constant matrix P:

```
MatrixForm[morph = Morphisms[R, R, {0, 0}, p, A]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

$$\begin{pmatrix} p[1][1] & 0 & p[3][1] & 0 \\ 0 & p[1][1] & 0 & p[3][1] \\ p[3][1] & 0 & p[1][1] & 0 \\ 0 & p[3][1] & 0 & p[1][1] \end{pmatrix}$$

From this P, let us try to compute idempotent endomorphisms of M:

```
MatrixForm /@ (PAll = IdempotentMorphisms[morph, R, p, A])
```

$$\left\{ \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} \frac{1}{2} & 0 & -\frac{1}{2} & 0 \\ 0 & \frac{1}{2} & 0 & -\frac{1}{2} \\ -\frac{1}{2} & 0 & \frac{1}{2} & 0 \\ 0 & -\frac{1}{2} & 0 & \frac{1}{2} \end{pmatrix}, \begin{pmatrix} \frac{1}{2} & 0 & \frac{1}{2} & 0 \\ 0 & \frac{1}{2} & 0 & \frac{1}{2} \\ \frac{1}{2} & 0 & \frac{1}{2} & 0 \\ 0 & \frac{1}{2} & 0 & \frac{1}{2} \end{pmatrix}, \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix} \right\}$$

Considering the nontrivial idempotent endomorphism PAll[[2]], we obtain the following decomposition:

```
MatrixForm[Decomposition[R, PAll[[2]], A]]
```

$$\begin{pmatrix} -i\,D_{x3} - D_{x4} & i\,D_{x1} - D_{x2} & 0 & 0 \\ i\,D_{x1} + D_{x2} & i\,D_{x3} - D_{x4} & 0 & 0 \\ 0 & 0 & i\,D_{x3} - D_{x4} & -i\,D_{x1} + D_{x2} \\ 0 & 0 & -i\,D_{x1} - D_{x2} & -i\,D_{x3} - D_{x4} \end{pmatrix}$$

Considering the nontrivial idempotent endomorphism PAll[[3]], we obtain the following decomposition:

```
MatrixForm[Decomposition[R, PAll[[3]], A]]
```

$$\begin{pmatrix} i D_{x3} - D_{x4} & -i D_{x1} + D_{x2} & 0 & 0 \\ -i D_{x1} - D_{x2} & -i D_{x3} - D_{x4} & 0 & 0 \\ 0 & 0 & -i D_{x3} - D_{x4} & i D_{x1} - D_{x2} \\ 0 & 0 & i D_{x1} + D_{x2} & i D_{x3} - D_{x4} \end{pmatrix}$$